JSAG **1** (2009), 1 – 4

The Journal of Software for
Algebra and Geometry

EdgeIdeals: a package for (hyper)graphs

CHRISTOPHER A. FRANCISCO, ANDREW HOEFEL, AND ADAM VAN TUYL

ABSTRACT. We introduce a *Macaulay 2* package, entitled *EdgeIdeals*, that allows one to experiment with graphs and hypergraphs via the edge ideal correspondence. At the core of our package are new classes for defining graphs and hypergraphs.

INTRODUCTION. The edge ideal of a (hyper)graph enables one to study (hyper)graphs using the tools of commutative algebra. The work of [1, 3, 6, 7, 9, 11, 12, 13, 14, 15], among others, has focused on building a dictionary between commutative algebra and graph theory. In this note we introduce a new package, entitled *EdgeIdeals*, written for *Macaulay 2* [5], which exploits this dictionary. The goal of this package is to provide a family of functions that will enable the user to experiment with simple graphs and hypergraphs within software specifically designed for commutative algebra and algebraic geometry, thus facilitating future research.

The underlying algorithms in this package use the notion of an edge ideal, a monomial ideal whose minimal generators correspond to the edges of the hypergraph. This algebraic construction necessitates that we work with clutters, hypergraphs in which no edge is a subset of another, since otherwise the edge of larger cardinality would not be a minimal generator of the edge ideal, and we would lose information.

We believe that some of the most useful parts of the package are methods to compute common invariants in (hyper)graph theory, such as `chromaticNumber` and `cliqueNumber`, methods to extract certain structures or test for particular features of graphs, such as `allOddHoles` and `isChordal`, and methods for computing random graphs and hypergraphs, such as `randomGraph` and `randomHyperGraph`, which allow the user to test conjectures efficiently. We also take advantage of the *SimplicialComplexes* package by Popescu, Smith, and Stillman to allow users to access the various simplicial complexes associated to a hypergraph. (See [10] for more on monomial ideals and *Macaulay 2*.)

As research on correspondences between (hyper)graphs and square-free monomial ideals continues, we envision including more algebraic methods for computing combinatorial invariants.

MATHEMATICAL BACKGROUND. For the purposes of this note and package, we define a **hypergraph** to be a pair $\mathscr{H} = (\mathscr{X}, \mathscr{E})$, where $\mathscr{X} = \{x_1, \dots, x_n\}$ is the set of vertices, and $\mathscr{E} = \{E_1, \dots, E_t\}$, a collection of subsets of $\mathscr{X}$, is the set of edges with the additional property that if $E_i \subseteq E_j$, then $i = j$. The standard usage of the term of hypergraph does not require the condition on the containment of edges. What we will call a hypergraph is sometimes called a **clutter** or **Sperner system**. If a hypergraph $\mathscr{H}$ has $|E_i| = 2$ for each $E_i \in \mathscr{E}$, then we usually call $\mathscr{H}$ a **simple graph**.

The package *EdgeIdeals* is based upon the correspondence between hypergraphs and square-free monomial ideals. The correspondence is defined as follows. Let $\mathscr{H} = (\mathscr{X}, \mathscr{E})$ be a hypergraph with

2000 *Mathematics Subject Classification.* 13F55, 05E99.
EdgeIdeals version 1.0.

vertex set $\mathscr{X} = \{x_1,\dots,x_n\}$. Fix a field k, and consider the polynomial ring $R = k[x_1,\dots,x_n]$. We identify each vertex of $\mathscr{H}$ with the corresponding variable in R. The edges of $\mathscr{H}$ are used to define an ideal, called the **edge ideal** of $\mathscr{H}$, in the ring R: $I(\mathscr{H}) = \langle\{\prod_{x\in E} x \mid E \in \mathscr{E}\}\rangle$. Conversely, any square-free monomial ideal I minimally generated by $\{m_1,\dots,m_s\}$ corresponds to a hypergraph, and the edge E_i corresponds to the variables in the support of the monomial m_i. Edge ideals are essentially equivalent to the notion of a **facet ideal** as first defined by Faridi [2] (see the discussion in [8]).

There is a delicate point in the above correspondence that revolves around isolated vertices. An isolated vertex is not considered an edge in a graph, and thus we omit it from the minimal generators in an edge ideal of a graph (though in our package, we do allow hypergraphs to have edges of cardinality one). One problem in handling graphs and hypergraphs is not knowing whether the user intended to have isolated vertices or whether he or she simply created a polynomial ring with extra variables, unsure in advance how many he or she would use. To address this, we created several alternate methods. For example, if the user forms the ring `R=ZZ/101[a,b,c]` and the graph `G` with edge set consisting of the single edge connecting `a` and `b`, `isConnectedGraph G` would return `false`, treating the vertex corresponding to `c` as an isolated vertex and its own connected component. In contrast, `isConnected G` would return `true`, assuming the user simply created a polynomial ring that was too large. See the documentation for more information about this and hypergraphs that have edges of cardinality one.

EXAMPLE. At the heart of the *EdgeIdeals* package are two new classes that are entitled `HyperGraph` and `Graph`. The `HyperGraph` class can only be used to represent hypergraphs. The class `Graph` extends from `HyperGraph` and inherits all of the methods of `HyperGraph`. Functions have been made that accept objects of either type as input.

In our example below, we will illustrate a theorem [15, Theorem 6.4.7] that says the independence complex of a Cohen-Macaulay bipartite graph has a simplicial shelling. We begin by inputting a graph and verifying the Cohen-Macaulay and bipartite properties.

```
i1 : loadPackage "EdgeIdeals";

i2 : R = QQ[x_1..x_3,y_1..y_3];

i3 : G = graph(R,{x_1*y_1,x_2*y_2,x_3*y_3, x_1*y_2,x_1*y_3,x_2*y_3})

o3 = Graph{edges => {{x , y }, {x , y }, {x , y }, {x , y }, {x , y }, {x , y }}}
                       1   1     2   2     3   3     1   2     1   3     2   3
           ring => R
           vertices => {x , x , x , y , y , y }
                          1   2   3   1   2   3

o3 : Graph

i4 : isCM G and isBipartite G

o4 = true
```

When defining a (hyper)graph, the user specifies the vertex set by defining a polynomial ring, while the edges are written as a list of square-free monomials (there are alternative ways of listing the edges). A (hyper)graph is stored as a hash table which contains the list of edges, the polynomial ring, and the list of vertices.

```
i5 : L = getGoodLeaf(G)

o5 = {x , y }
       1   1
```

```
o5 : List
i6 : degreeVertex(G,y_1)
o6 = 1
i7 : H = inducedHyperGraph(G, vertices(G) - set(L))
o7 = HyperGraph{edges => {{x , y }, {x , y }, {x , y }}}
                            2   2     3   3     2   3
                ring => QQ [x , x , y , y ]
                             2   3   2   3
                vertices => {x , x , y , y }
                              2   3   2   3
o7 : HyperGraph
```

A Cohen-Macaulay bipartite graph must contain a leaf, which we retrieve above. We remove the leaf, to form the induced graph, and at the same time, we identify the vertex of degree one in the leaf.

```
i8 : K = simplicialComplexToHyperGraph independenceComplex H;
i9 : edges K
o9 = {{x , x }, {x , y }, {y , y }}
        2   3     3   2     2   3
o9 : List
```

Above, we formed the independence complex of H, that is, the simplicial complex whose facets correspond to the maximal independent sets of H. We then change the type from a simplicial complex to a hypergraph, which we call K. Notice that these edges give a shelling.

```
i10 : use ring K;
i11 : A = apply(edges(K), e -> append(e, y_1));
i12 : B = apply(edges inducedHyperGraph(K, {x_2,x_3}), e-> append(e, x_1));
i13 : shelling = join(A,B)
o13 = {{x , x , y }, {x , y , y }, {y , y , y }, {x , x , x }}
         2   3   1     3   2   1     2   3   1     2   3   1
o13 : List
i14 : independenceComplex(G)
o14 = | y_1y_2y_3 x_3y_1y_2 x_2x_3y_1 x_1x_2x_3 |
o14 : SimplicialComplex
```

Using the method found in the proof of [15, Theorem 6.4.7], we now can form a shelling of the original independence complex. Notice that our shelling is a permutation of the facets of the independence complex defined from G.

REFERENCES.

[1] D. Eisenbud, M. Green, K. Hulek and S. Popescu, Restricting linear syzygies: algebra and geometry, *Compositio Math.* **141** (2005), no. 6, 1460 – 1478.

[2] S. Faridi, The facet ideal of a simplicial complex. *Manuscripta Math.* **109** (2002), no. 2, 159 – 174.

[3] C.A. Francisco, H.T. Hà, Whiskers and sequentially Cohen-Macaulay graphs. *J. Combin. Theory Ser. A* **115** (2008), no. 2, 304 – 316.

[4] C.A. Francisco, H.T. Hà, A. Van Tuyl, Algebraic methods for detecting odd holes in a graph, arXiv:math.AC/0806.1159v1.

[5] D. R. Grayson and M. E. Stillman, *Macaulay 2*, a software system for research in algebraic geometry, available at www.math.uiuc.edu/Macaulay2/.

[6] H.T. Hà, A. Van Tuyl, Resolutions of square-free monomial ideals via facet ideals: a survey, *Contemp. Math.* **448** (2007), 91 – 117.

[7] H.T. Hà, A. Van Tuyl, Splittable ideals and the resolutions of monomial ideals, *J. Algebra* **309** (2007), no. 1, 405 – 425.

[8] H.T. Hà, A. Van Tuyl, Monomial ideals, edge ideals of hypergraphs, and their minimal graded free resolutions, *J. Algebraic Combin.* **27** (2008), no. 2, 215 – 245.

[9] J. Herzog, T. Hibi, and X. Zheng, Cohen-Macaulay chordal graphs, *J. Combin. Theory Ser. A* **113** (2006), no. 5, 911 – 916.

[10] S. Hosten, G. Smith, Monomial ideals. In *Computations in algebraic geometry with Macaulay 2*, Algorithms and Computations in Mathematics **8** (2001), Springer-Verlag, 73 – 100.

[11] S. Jacques, Betti numbers of graph ideals. Ph.D. Thesis, University of Sheffield (2004), arXiv:math.AC/0410107.

[12] A. Simis, W.V. Vasconcelos, R.H. Villarreal, On the ideal theory of graphs, *J. Algebra* **167** (1994), no. 2, 389 – 416.

[13] B. Sturmfels, S. Sullivant, Combinatorial secant varieties. *Pure Appl. Math. Q.* **2** (2006), no. 3, part 1, 867 – 891.

[14] R.H. Villarreal, Cohen-Macaulay graphs. *Manuscripta Math.* **66** (1990), no. 3, 277 – 293.

[15] R.H. Villarreal, *Monomial algebras.* Monographs and Textbooks in Pure and Applied Mathematics, **238**. Marcel Dekker, Inc., New York, 2001.

[16] X. Zheng, Homological properties of monomial ideals associated to quasi-trees and lattices. Ph.D. Thesis, Essen (2004).

RECEIVED : 2008-09-24 REVISED : 2009-06-08 ACCEPTED : 2009-06-27

chris@math.okstate.edu : Department of Mathematics, Oklahoma State University, 401 Mathematical Sciences, Stillwater, OK 74078, USA.

andrew.hoefel@mathstat.dal.ca : Department of Mathematics & Statistics, Chase Building, Dalhousie University, Halifax, NS B3H 3J5, Canada.

avantuyl@lakeheadu.ca : Department of Mathematical Sciences, Lakehead University, Thunder Bay, ON P7B 5E1, Canada.